

# Formalized Haar Measure

Floris van Doorn  

University of Pittsburgh, USA

---

## Abstract

We describe the formalization of the existence and uniqueness of the Haar measure in the Lean theorem prover. The Haar measure is an invariant regular measure on locally compact groups, and it has not been formalized in a proof assistant before. We will also discuss the measure theory library in Lean’s mathematical library `mathlib`, and discuss the construction of product measures and the proof of Fubini’s theorem for the Bochner integral.

**2012 ACM Subject Classification** Theory of computation → Type theory; Mathematics of computing → Mathematical analysis

**Keywords and phrases** Haar measure, measure theory, Bochner integral, Lean, interactive theorem proving, formalized mathematics

**Digital Object Identifier** 10.4230/LIPIcs.ITP.2021.24

**Related Version** *Full Version*: <https://arxiv.org/abs/2102.07636>

**Supplementary Material** *Software (Source Code)*: [https://github.com/leanprover-community/mathlib/tree/master/src/measure\\_theory](https://github.com/leanprover-community/mathlib/tree/master/src/measure_theory)  
archived at `swh:1:dir:4b232212fa94778af7c966340636fd0e061d9598`

**Funding** This paper is supported by the Sloan Foundation (grant G-2018-10067)

**Acknowledgements** I want to thank Jeremy Avigad, Tom Hales, the Leanprover community and the anonymous reviewers for proofreading this paper.

## 1 Introduction

Measure theory is an important part of mathematics, providing a rigorous basis for integration and probability theory. The main object of study in this part is the concept of a measure, which assigns a size to the measurable subsets of the space in question. The most widely used measure is the Lebesgue measure  $\lambda$  on  $\mathbb{R}$  (or  $\mathbb{R}^n$ ) with the property that  $\lambda([a, b]) = b - a$ . The Lebesgue measure is *translation invariant*, meaning that translating a measurable set does not change its measure.

An important generalization of the Lebesgue measure is the Haar measure, which provides a measure on locally compact groups. This measure is also invariant under translations, i.e. applying the group operation. For non-abelian groups we have to distinguish between left and right Haar measures, which are invariant when applying the group operation on the left and right, respectively. A Haar measure is essentially unique, which means that any invariant measure is a constant multiple of a chosen Haar measure. The Haar measure is vital in various areas of mathematics, including harmonic analysis, representation theory, probability theory and ergodic theory.

In this paper we describe the formal definition of the Haar measure in the Lean theorem prover [5], and prove its uniqueness. We build on top of the Lean mathematical library `mathlib` [4]. We heavily use the library for measure theory in `mathlib`, improving it in the process. As part of this formalization we expanded the library with product measures and Fubini’s theorem, along with many other contributions to existing libraries. The results in this paper have been fully integrated into `mathlib`.



© Floris van Doorn;

licensed under Creative Commons License CC-BY 4.0

12th International Conference on Interactive Theorem Proving (ITP 2021).

Editors: Liron Cohen and Cezary Kaliszyk; Article No. 24; pp. 24:1–24:17

Leibniz International Proceedings in Informatics





LIPICs Schloss Dagstuhl – Leibniz-Zentrum für Informatik, Dagstuhl Publishing, Germany

Many proof assistants have a library of measure theory and integration, including Isabelle [17, 12], HOL/HOL4 [13, 2, 15], HOL Light [11], Mizar [7], PVS [14] and Coq [19]. Most of these libraries focus on the Lebesgue measure and the Lebesgue integral, and on applications to probability theory or analysis in Euclidean spaces.

The Isabelle/HOL library has an especially good development of measure theory, including product measures and the Bochner integral [1]. The Lean library also has these concepts, and the development of them was heavily inspired by the Isabelle/HOL library (and partially written by Johannes Hölzl, who also wrote parts of the Isabelle/HOL library). However, to my knowledge, the Haar measure has not been formalized in any proof assistant other than Lean.

In this paper we will discuss the measure theory library in Lean (in Section 3), which was already in place before this formalization started, and was developed by Johannes Hölzl, Mario Carneiro, Zhouhang Zhou and Yury Kudryashov, among others. The other sections describe contributions that are new as of this formalization: product measures (in Section 4), the definition of the Haar measure (in Section 5), and the uniqueness of the Haar measure (in Section 6).

We will link to specific results in `mathlib` using the icon . To ensure that these links will work while `mathlib` is updated, we link to the version of `mathlib` as of writing this paper. Readers of this paper are also encouraged to use the search function of the `mathlib` documentation pages  but we do not link to specific pages of the documentation, as these links are prone to break as the library evolves.

We used different sources for this formalization. We started out using the notes by Jonathan Gleason [8] that gave a construction of the Haar measure in detail. However, one of the main claims in that paper was incorrect, as described in Section 5. We then used the books by Halmos and Cohn [10, 3] to fill in various details. For product measures we followed Cohn [3]. For Fubini’s theorem for the Bochner integral we followed the formalization in Isabelle. We couldn’t find a good source for this proof in the literature, so we will provide detailed proofs of this result in Section 4. A different proof can be found in [16]. For the uniqueness of the Haar measure we followed Halmos [10]. While we define the Haar measure for any locally compact group, we prove the uniqueness only for second-countable locally compact groups. The reason is that Halmos gives a uniqueness proof using measurable groups, and only second-countable topological groups form measurable groups. In the proof of a key theorem of Halmos there was a gap, which we fixed in the formalization by changing the statement of that theorem (see Section 6).

## 2 Preliminaries

### 2.1 Lean and `mathlib`

Lean [5] is a dependently typed proof assistant with a strong metaprogramming framework [6]. Its logic contains dependent function types denoted  $\prod i : \iota, X i$ , universes `Type*` (the star denotes an arbitrary universe level) and inductive types. It has a large community-driven mathematical library, called `mathlib`. We use results from various areas in this library, including topology, analysis and algebra. In particular the existing library for topological groups was very helpful for this formalization. The fact that Lean provides a very expressive type theory was convenient in the definition of the Bochner integral, as we will discuss in Section 3.

`mathlib` uses *type-classes* [20, 18] to organize the mathematical structures associated to various types, using the partially bundled approach. For example, when  $G$  is a type, the

statement `[group G]` states that `G` has the structure of a group, and `[topological_space G]` states that `G` has the structure of a topological space. The square brackets mean that the argument will be implicitly inserted in definitions and theorems using type-class inference. Some type-classes are *mixins*, and take other type-classes as arguments. For example, the class `topological_group` takes as (implicit) argument the structure of a group and a `topological_space`. This means that we need to write `[group G] [topological_space G] [topological_group G]` to state that `G` is a topological group. Similarly, predicates on topological spaces — such as `t2_space`, `locally_compact_space` or `compact_space` — take the topological space structure as argument. This design decision makes it somewhat verbose to get the right structure on a type, but has the advantage that these components can be freely mixed as needed.

Lean distinguishes groups written multiplicatively (`group`) from groups written additively (`add_group`). This is convenient since it automatically establishes multiplicative notation ( $1 * x^{-1}$ ) or additive notation ( $0 + -x$  or  $0 - x$ ) when declaring a group structure. Also, most theorems exist for both versions; the property  $x * y = y * x$  is called `mul_comm` and the property  $x + y = y + x$  is `add_comm`. To avoid code duplication, there is an attribute `@[to_additive]` that, when applied to a definition or theorem for multiplicative groups, automatically generates the corresponding declaration for additive groups by replacing all multiplicative notions with the corresponding additive ones.

Some code snippets in this paper have been edited slightly for the sake of readability.

## 2.2 Mathematics in mathlib

The notation for the image of a set  $A : \text{set } X$  under a function  $f : X \rightarrow Y$  is  $f '' A = \{y : Y \mid \exists x \in A, f x = y\}$  and the preimage of  $B : \text{set } Y$  is  $f^{-1} B = \{x : X \mid f x \in B\}$ . The notation  $\lambda x, f x$  is used for the function  $x \mapsto f(x)$ . We write  $(x : X)$  for the element  $x$  with its type  $X$  given explicitly. This can also be used to *coerce*  $x$  to the type  $X$ .

`mathlib` has a large topology library, where many concepts are defined in terms of filters. The following code snippet contains some notions in topology used in this paper.

```
class t2_space (X : Type*) [topological_space X] : Prop :=
  (t2 : ∀ x y, x ≠ y → ∃ U V : set X, is_open U ∧ is_open V ∧
    x ∈ U ∧ y ∈ V ∧ U ∩ V = ∅)

def is_compact {X : Type*} [topological_space X] (K : set X) :=
  ∀ {f : filter X} [ne_bot f], f ≤ ℘ K → ∃ x ∈ K, cluster_pt x f

class locally_compact_space (X : Type*) [topological_space X] : Prop :=
  (local_compact_nhds : ∀ (x : X) (U ∈ ℱ x), ∃ K ∈ ℱ x, K ⊆ U ∧
    is_compact K)

class second_countable_topology (X : Type*) [t : topological_space X] :
  Prop :=
  (is_open_generated_countable : ∃ C : set (set X), countable C ∧
    t = generate_from C)

class complete_space (X : Type*) [uniform_space X] : Prop :=
  (complete : ∀ {f : filter X}, cauchy f → ∃ x, f ≤ ℱ x)

class topological_group (G : Type*) [topological_space G] [group G] :=
  (continuous_mul : continuous (λ p : G × G, p.1 * p.2))
  (continuous_inv : continuous (inv : G → G))
```

## 24:4 Formalized Haar Measure

Some of these definitions might look unfamiliar, formulated in terms of filters, using in particular the principal filter  $\mathcal{P}$  and the neighborhood filter  $\mathcal{N}$ . For example, the second definition states that a set  $K$  is *compact* if any filter  $\mathfrak{f} \neq \perp$  (i.e.  $\mathfrak{f}$  doesn't contain every subset of  $X$ ) containing  $K$  has a cluster point  $x$  in  $K$ , which means that  $\mathcal{N} x \sqcap \mathfrak{f} \neq \perp$ . It is equivalent to the usual notion of compactness that every open cover of  $K$  has a finite subcover. A complete space is a space where every Cauchy sequence converges, which is formulated in the general setting of *uniform spaces*, which simultaneously generalize metric spaces and topological groups. We define a *locally compact group* to be a topological group in which the topology is both locally compact and Hausdorff ( $T_2$ ).

Lean's notion of infinite sums is defined for absolutely convergent series as the limit of all finite partial sums:

```
def has_sum {I M : Type*} [add_comm_monoid M] [topological_space M]
  (f : I → M) (x : M) : Prop :=
  tendsto (λ s : finset I, ∑ i in s, f i) at_top (N x)
```

If  $M$  is Hausdorff, then the sum is unique if it exists, and we denote it by  $\sum' i, f i$ . If the series does not have a sum, it is defined to be 0.

The type `ennreal` is the type of nonnegative real numbers extended by a new element  $\infty$ , which we will also denote  $[0, \infty]$  in this paper.

A *normed space*  $X$  is a vector space over a normed field  $F$  equipped with a norm  $\|\cdot\| : X \rightarrow \mathbb{R}$  satisfying

- $(X, d)$  is a metric space, with the distance function given by  $d(x, y) := \|x - y\|$ .
- for  $r \in F$  and  $x \in X$  we have  $\|r \cdot x\| = \|r\| \cdot \|x\|$ .

In this paper, we will only consider normed spaces over  $\mathbb{R}$ , which is a normed field with its norm given by the absolute value. A *Banach space* is a complete normed space. In Lean we do not explicitly define Banach spaces, but instead talk about complete normed spaces.

### 3 Measure Theory in mathlib

In this section we describe the background measure theory library used in this formalization.

A basic notion in measure theory is a  $\sigma$ -*algebra* on  $X$ , which is a nonempty collection of subsets of  $X$  that is closed under complements and countable unions. In `mathlib` this is formulated as a type-class, with name `measurable_space`:

```
class measurable_space (X : Type*) :=
  (is_measurable : set X → Prop)
  (is_measurable_empty : is_measurable ∅)
  (is_measurable_compl : ∀ A, is_measurable A → is_measurable Aᶜ)
  (is_measurable_Union : ∀ A : ℕ → set X, (∀ i, is_measurable (A i)) →
    is_measurable (⋃ i, A i))
```

We say that a function is *measurable* if the preimage of every measurable set is measurable:

```
def measurable [measurable_space X] [measurable_space Y] (f : X → Y) :=
  ∀ {B : set Y}, is_measurable B → is_measurable (f ⁻¹ B)
```

Note that we use `is_measurable` for sets and `measurable` for functions.

In `mathlib`, measures are defined as special case of outer measures. An outer measure on  $X$  is a monotone function  $m : \mathcal{P}(X) \rightarrow [0, \infty]$  such that  $m(\emptyset) = 0$  and  $m$  is countably subadditive:

```

structure outer_measure (X : Type*) :=
  (measure_of : set X → ennreal)
  (empty : measure_of ∅ = 0)
  (mono : ∀ {A₁ A₂}, A₁ ⊆ A₂ → measure_of A₁ ≤ measure_of A₂)
  (Union_nat : ∀ (A : ℕ → set X),
    measure_of (⋃ i, A i) ≤ (∑' i, measure_of (A i)))

```

A measure on a measurable space is an outer measure with two extra properties: it is countably additive on measurable sets, and given the value on measurable sets, the outer measure is the maximal one that is compatible with these values (which is called `trim` in the snippet below).  $\square$

```

structure measure (X : Type*) [measurable_space X]
  extends outer_measure X :=
  (m_Union : ∀ {A : ℕ → set X},
    (∀ i, is_measurable (A i)) → pairwise (disjoint on A) →
    measure_of (⋃ i, A i) = ∑' i, measure_of (A i))
  (trimmed : to_outer_measure.trim = to_outer_measure)

```

This definition has two very convenient properties:

1. We can apply measures to any set, without having to provide a proof that the set is measurable.
2. Two measures are equal when they are equal on measurable sets.  $\square$

Given a measure  $\mu$  on  $X$  and a measurable map  $f : X \rightarrow Y$ , we can define the *pushforward*  $f_*\mu$ ,  $\square$  which is a measure on  $Y$ , defined on measurable sets  $A$  by  $(f_*\mu)(A) = \mu(f^{-1}(A))$ .  $\square$

We proceed to define lower Lebesgue integration  $\int^-$  for functions  $f : X \rightarrow [0, \infty]$ , where  $X$  is a measurable space. We annotate this integral with a minus sign to distinguish it from the Bochner integral. This is done first for *simple functions*, i.e. functions with finite range with the property that the preimage of all singletons are measurable. The integral of a simple function  $g$  is simply  $\square$

$$\int^- g \, d\mu = \int^- g(x) \, d\mu(x) = \sum_{y \in g(X)} \mu(g^{-1}\{y\}) \cdot y \in [0, \infty].$$

If  $f : X \rightarrow [0, \infty]$  is any function, we can define the (*lower*) *Lebesgue integral* of  $f$  to be the supremum of  $\int^- g \, d\mu(x)$  for all simple functions  $g$  that satisfy  $g \leq f$  (pointwise).  $\square$  In `mathlib` we denote the Lebesgue integral of  $\mathbf{f}$  by  $\int^- \mathbf{x}, \mathbf{f} \, \mathbf{x} \, \partial\mu$ . We prove the standard properties of the Lebesgue integral for nonnegative functions, such as the monotone convergence theorem:  $\square$

```

theorem lintegral_supr {f : ℕ → X → ennreal}
  (hf : ∀ n, measurable (f n)) (h_mono : monotone f) :
  (∫^- x, ⋂ n, f n x ∂μ) = (⋂ n, ∫^- x, f n x ∂μ)

```

After defining the Lebesgue integral for nonnegative functions, most books and formal libraries continue to define the Lebesgue integral for (signed) functions  $f : X \rightarrow \mathbb{R}$ . This is defined as the difference of the integrals of the positive and negative parts of  $f$ . However, this is not very general: separate definitions are needed for integrals for functions with codomain  $\mathbb{C}$  or  $\mathbb{R}^n$ . Instead, we opt for the more general definition of the *Bochner integral*, which is defined for functions that map into any second-countable real Banach space.

If  $X$  is a topological space, we equip it with the *Borel*  $\sigma$ -algebra, which is the smallest  $\sigma$ -algebra that contains all open sets (or equivalently, all closed sets).  $\square$  The fact that  $X$  is

## 24:6 Formalized Haar Measure

equipped with the Borel  $\sigma$ -algebra is a mixin, and is specified by providing the arguments [topological\_space X] [measurable\_space X] [borel\_space X].

Let  $X$  be a measurable space and  $E$  be a second-countable real Banach space. We define two quotients on functions from  $X$  to  $E$ .

Let  $X \rightarrow_{\mu} E$  be the measurable maps  $X \rightarrow E$  modulo  $\mu$ -a.e. (almost everywhere) equivalence, i.e. modulo the relation

$$f =_{\mu} g \iff \mu(\{x \mid f(x) \neq g(x)\}) = 0. \quad \square$$

Note that we are using the fact that Lean is based on dependent type theory here, since  $X \rightarrow_{\mu} E$  is a type that depends on  $\mu$ . Here our formalization differs from the one in Isabelle, since in Isabelle this type cannot be formed. Instead, a similar argument is given purely on functions, without passing to the quotient.

If  $f : X \rightarrow E$  we say that  $f$  is *integrable* if  $\square$

- $f$  is  $\mu$ -a.e. measurable, i.e. there is a measurable function  $g : X \rightarrow E$  with  $f =_{\mu} g$ ,<sup>1</sup> and
- $\int^{-} \|f(x)\| d\mu(x) < \infty$ .

Being integrable is preserved by  $\mu$ -a.e. equivalence, and we define the  $L^1$ -space  $L^1(X, \mu; E)$  as those equivalence classes of functions in  $X \rightarrow_{\mu} E$  that are integrable.  $\square$   $L^1(X, \mu; E)$  is a normed space,  $\square$  with the norm given by

$$\|f\| := \int^{-} \|f(x)\| d\mu(x).$$

We define the Bochner integral first for simple functions, similar to the definition for the Lebesgue integral. If  $g : X \rightarrow E$  is a simple function then its Bochner integral is  $\square$

$$\int g d\mu = \int g(x) d\mu(x) = \sum_{\substack{y \in g(X) \text{ s.t.} \\ \mu(g^{-1}\{y\}) < \infty}} \mu(g^{-1}\{y\}) \cdot y \in E.$$

The symbol  $\cdot$  denotes the scalar multiplication in the Banach space  $E$ . Since this definition respects  $\mu$ -a.e. equivalence, we can also define the Bochner integral on the simple  $L^1$  functions.  $\square$  On the simple  $L^1$  functions this definition is continuous  $\square$  and the simple  $L^1$  functions are dense in all  $L^1$  functions,  $\square$  and  $E$  is complete, so we can extend the Bochner integral to all  $L^1$  functions.  $\square$

Finally, for an arbitrary function  $f : X \rightarrow E$  we define its Bochner integral to be 0 if  $f$  is not integrable and the integral of  $[f] \in L^1(X, \mu; E)$  otherwise, where  $[f]$  is the equivalence class of  $f$ .  $\square$  The two notions of integrals agree on integrable functions  $f : X \rightarrow \mathbb{R}_{\geq 0}$ :  $\square$

$$\int^{-} f d\mu = \int f d\mu.$$

However, when  $f$  has infinite Lebesgue integral, then the equality does not hold: the LHS is  $\infty$ , while the RHS is 0. We then prove useful properties about the Bochner integral, such as the dominated convergence theorem:  $\square$

<sup>1</sup> Until recently, the definition of “integrable” in `mathlib` required  $f$  to be measurable. The definition was modified to increase generality. The motivating reason was that we would regularly take integrals of a function  $f$  with a measure  $\mu|_A$  defined by  $\mu|_A(B) = \mu(A \cap B)$ . A function that is (for example) continuous on  $A$  is also  $\mu|_A$ -a.e. measurable, and we usually do not need any information about the behavior of  $f$  outside  $A$ . In the previous definition of “integrable” we required  $f$  to be measurable everywhere, which would be an unnecessary additional assumption.

```

theorem tendsto_integral_of_dominated_convergence
  {F : ℕ → X → E} {f : X → E} (bound : X → ℝ)
  (F_measurable : ∀ n, ae_measurable (F n) μ)
  (f_measurable : ae_measurable f μ)
  (bound_integrable : integrable bound μ)
  (h_bound : ∀ n, ∀m x ∂μ, ||F n x|| ≤ bound x)
  (h_lim : ∀m x ∂μ, tendsto (λ n, F n x) at_top (ℕ (f x))) :
  tendsto (λ n, ∫ x, F n x ∂μ) at_top (ℕ (∫ x, f x ∂μ))

```

In this statement,  $\forall^m x \partial\mu$ ,  $P x$  means that  $P x$  holds for  $\mu$ -almost all  $x$ , in other words that  $\mu \{x \mid \neg P x\} = 0$ , and  $\text{tendsto } (\lambda n, g n) \text{ at\_top } (\mathcal{N} x)$  means that  $g n$  tends to  $x$  as  $n$  tends to  $\infty$ .

Other results about integration include part 1 [↗](#) and 2 [↗](#) of the fundamental theorem of calculus, but we will not use these results in this paper.

## 4 Products of Measures

To prove that the Haar measure is essentially unique, we need to work with product measures and iterated integrals. In this section we will define the product measure and prove Tonelli's and Fubini's theorem. (Sometimes both theorems are called Fubini's theorem, but in this paper we will distinguish them.) Tonelli's theorem characterizes the Lebesgue integral for nonnegative functions in the product space, and Fubini's theorem does the same (under stronger conditions) for the Bochner integral. By symmetry, both theorems also give sufficient conditions for swapping the order of integration when working with iterated integrals.

The contents of this section have been formalized before in Isabelle. Product measures, Tonelli's theorem and Fubini's theorem for the Lebesgue integral are discussed in the paper [12]. Fubini's theorem for the Bochner integral is significantly harder to prove, but has also been formalized in Isabelle after the appearance of that paper. [↗](#) We read through the proof in the Isabelle formalization for key ideas in various intermediate lemmas.

In this section we will restrict our attention to  $\sigma$ -finite measures, since product measures are much nicer for  $\sigma$ -finite measures, and most of the results do not hold without this condition. We say that a measure  $\mu$  on a space  $X$  is  $\sigma$ -finite if  $X$  can be written as a union of countably many subsets  $(A_i)_i$  with  $\mu(A_i) < \infty$  for all  $i$ . [↗](#)

First we need to know that limits of measurable functions are measurable.

► **Lemma 1.** *Let  $X$  be a measurable space.*

1. *Given a sequence of measurable functions  $f_n : X \rightarrow [0, \infty]$  for  $n \in \mathbb{N}$ , the pointwise suprema and infima  $f(x) := \sup\{f_n(x)\}_n$  and  $g(x) := \inf\{f_n(x)\}_n$  are measurable. [↗](#)*
2. *If a sequence of measurable functions  $f_n : X \rightarrow [0, \infty]$  for  $n \in \mathbb{N}$  converges pointwise to a function  $f$ , then  $f$  is measurable. [↗](#)*
3. *Let  $Y$  be a metric space equipped with the Borel  $\sigma$ -algebra. If a sequence of measurable functions  $f_n : X \rightarrow Y$  for  $n \in \mathbb{N}$  converges pointwise to a function  $f$ , then  $f$  is measurable. [↗](#)*

**Proof.** For part 1, note that the collection of intervals  $[0, x)$  generate the  $\sigma$ -algebra on  $[0, \infty]$ , as do the intervals  $(x, \infty]$ . The conclusion follows from the fact that measurable sets are closed under countable unions and observing that  $f^{-1}([0, x)) = \bigcup_n f_n^{-1}([0, x))$  and  $g^{-1}((x, \infty]) = \bigcup_n f_n^{-1}((x, \infty])$ .

Part 2 follows by observing that  $f(x) = \liminf_n f_n(x) = \sup_n \inf_{m \geq n} f_m(x)$ , and applying both claims in part 1.



## 24:8 Formalized Haar Measure

For part 3, let  $C$  be a closed set in  $Y$ . The function  $d(-, C) : Y \rightarrow [0, \infty]$  that assigns to each point the (minimal) distance to  $C$  is continuous. Therefore, we know that  $h(x) := d(f(x), C)$  is the limit of the measurable functions  $d(f_n(x), C)$  as  $n \rightarrow \infty$ . By part 2 we know that  $h$  is measurable. Therefore  $f^{-1}(C) = h^{-1}(\{0\})$  is measurable (the equality holds because  $C$  is closed).  $\blacktriangleleft$

If  $X$  and  $Y$  are measurable spaces, the  $\sigma$ -algebra on  $X \times Y$  is the smallest  $\sigma$ -algebra that makes the projection  $X \times Y \rightarrow X$  and  $X \times Y \rightarrow Y$  measurable.  $\square$  Alternatively, it can be characterized as the  $\sigma$ -algebra generated by sets of the form  $A \times B$  for measurable  $A \subseteq X$  and  $B \subseteq Y$ .  $\square$

For the rest of this section,  $\mu$  is a  $\sigma$ -finite measure on  $X$  and  $\nu$  is a  $\sigma$ -finite measure on  $Y$ . For a set  $A \subseteq X \times Y$  we write  $A_x = \{y \in Y \mid (x, y) \in A\}$  for a slice of  $Y$ . In Lean, this is written as `prod.mk x ⁻¹' A`. We want to define the product measure evaluated at  $A$  as an integral over the volume of the slices of  $A$ . To make sure that this makes sense, we need to prove that this is a measurable function.  $\square$

```
lemma measurable_measure_prod_mk_left [sigma_finite ν] {A : set (X × Y)}
  (hA : is_measurable A) : measurable (λ x, ν (prod.mk x ⁻¹' A))
```

This lemma crucially depends on the fact that  $\nu$  is  $\sigma$ -finite, but we omit the proof here. We can now define the product measure  $\mu \times \nu$  on  $X \times Y$  as  $\square$

$$(\mu \times \nu)(A) = \int^- \nu(A_x) d\mu(x). \quad (1)$$

It is not hard to see that  $(\mu \times \nu)(A \times B) = \mu(A)\nu(B)$  for measurable  $A \subseteq X$  and  $B \subseteq Y$   $\square$  and that  $\mu \times \nu$  is itself  $\sigma$ -finite.  $\square$  When we pushforward the measure  $\mu \times \nu$  across the map  $X \times Y \rightarrow Y \times X$  that swaps the coordinates, we get the measure  $\nu \times \mu$   $\square$  (this equality is easily checked because both measures are equal on rectangles). This immediately gives the symmetric version of (1), using the notation  $A^y = \{x \in X \mid (x, y) \in A\}$ :  $\square$

$$(\mu \times \nu)(A) = \int^- \mu(A^y) d\nu(y).$$

For a function  $f : X \times Y \rightarrow Z$  define  $f_x : Y \rightarrow Z$  by  $f_x(y) := f(x, y)$ . If  $Z = [0, \infty]$  let  $I_f^- : X \rightarrow [0, \infty]$  be defined by  $I_f^-(x) := \int_Y^- f_x d\nu = \int_Y^- f(x, y) d\nu(y)$ .

We can now formulate Tonelli's theorem.

► **Theorem 2** (Tonelli's theorem). *Let  $f : X \times Y \rightarrow [0, \infty]$  be a measurable function. Then  $\square \square$*

$$\int_{X \times Y}^- f d(\mu \times \nu) = \int_X^- \int_Y^- f(x, y) d\nu(y) d\mu(x) = \int_Y^- \int_X^- f(x, y) d\mu(x) d\nu(y),$$

and all the functions in the integrals above are measurable.  $\square \square$

**Proof.** We check the first equality and the measurability of  $I_f^-$ . The other claims follow by symmetry. For both these statements we use *induction on measurable functions into  $[0, \infty]$* ,  $\square$  which states that to prove a statement  $P(f)$  for all measurable functions  $f$  into  $[0, \infty]$ , it is sufficient to show

1. if  $\chi_A$  is the characteristic function of a measurable set and  $c \in [0, \infty]$ , we have  $P(c\chi_A)$ ;
2. if  $f_1$  and  $f_2$  are measurable functions with  $P(f_1)$  and  $P(f_2)$ , then  $P(f_1 + f_2)$ ;
3. if  $(f_i)_i$  is a monotone sequence of measurable functions such that  $P(f_i)$  for all  $i$ , we have  $P(\sup_i f_i)$ .



We first prove that  $I_f^-$  is measurable by induction on  $f$ .

In the induction base, if  $f = c\chi_A$ , then  $\int_Y^- f(x, y) \, d\nu(y) = c\nu(A_x)$ , so the measurability follows from `measurable_measure_prod_mk_left`. The two induction steps follow from the additivity of the integral and the monotone convergence theorem.

We now prove the first equality in Theorem 2, also by induction on  $f$ . In the induction base, if  $f = c\chi_A$ , then it is not hard to see that both sides reduce to  $c(\mu \times \nu)(A)$ . The first induction step follows by applying additivity thrice, and the second induction step follows by applying the monotone convergence theorem thrice. ◀

Fubini's theorem (sometimes called Fubini–Tonelli's theorem) states something similar for the Bochner integral. However, it is a bit harder to state and much harder to prove. For a function  $f : X \times Y \rightarrow E$  define  $I_f : X \rightarrow E$  by  $I_f(x) := \int_Y f_x \, d\nu = \int_Y f(x, y) \, d\nu(y)$ .

► **Theorem 3** (Fubini's theorem for the Bochner integral). *Let  $E$  be a second-countable Banach space and  $f : X \times Y \rightarrow E$  be a function.*

1. *If  $f$  is  $\mu \times \nu$ -a.e. measurable then  $f$  is integrable iff the following two conditions hold:*
  - *for almost all  $x \in X$  the function  $f_x$  is  $\nu$ -integrable;*
  - *the function  $I_{\|f\|}$  is  $\mu$ -integrable.*
2. *If  $f$  is integrable, then* ◻ ◻

$$\int_{X \times Y} f \, d(\mu \times \nu) = \int_X \int_Y f(x, y) \, d\nu(y) \, d\mu(x) = \int_Y \int_X f(x, y) \, d\mu(x) \, d\nu(y).$$

Moreover, all the functions in the integrals above are a.e. measurable. ◻ ◻

Note that Part 1 also has a symmetric counterpart, ◻ which we omit here. Note that the middle term in Part 2 is just  $\int_X I_f \, d\mu(x)$ .

**Proof. Measurability.** First suppose that  $f$  is measurable. In this case, we show that the function  $I_f$  is measurable. We can approximate any measurable function  $f$  into  $E$  by a sequence  $(s_n)_n$  of simple functions ◻ such that for all  $z$  we have  $\|s_n(z)\| \leq 2\|f(z)\|$  ◻ and  $s_n(z) \rightarrow f(z)$  as  $n \rightarrow \infty$ . ◻ Now define  $g_n : X \rightarrow E$  by

$$g_n(x) = \begin{cases} I_{s_n}(x) & \text{if } f_x \text{ is } \nu\text{-integrable} \\ 0 & \text{otherwise.} \end{cases}$$

Note that  $g_n$  is similar to  $I_{s_n}$ , except that we set it to 0 whenever  $f_x$  is not integrable. This is required to ensure that  $g_n$  converges to  $I_f$  (see below). We first check that  $g_n$  is measurable. Note that the set

$$\{x \mid f_x \text{ is } \nu\text{-integrable}\} = \{x \mid \int_Y \|f(x, y)\| \, d\nu(y) < \infty\}$$

is measurable, using Theorem 2. Therefore, to show that  $g_n$  is measurable, it suffices to show that  $I_{s_n}$  is measurable, which is true since it is a finite sum of functions of the form  $x \mapsto c\nu(A_x)$  (using that  $s_n$  is simple).

Secondly, we check that  $g_n$  converges to  $I_f$  pointwise. Let  $x \in X$ . If  $f_x$  is not integrable, it is trivially true, so assume that  $f_x$  is integrable. Then  $(s_n)_x$  is integrable for all  $n$ , and by the dominated convergence theorem  $I_{s_n}(x)$  converges to  $I_f(x)$ .

We conclude that  $I_f$  is measurable by Lemma 1.

Finally, if  $f$  is  $\mu \times \nu$ -a.e. measurable, then it is not hard to show that  $I_f$  is  $\mu$ -a.e. measurable.

**Integrability.** We now prove Part 1, so suppose that  $f$  is  $\mu \times \nu$ -a.e. measurable. By Tonelli's theorem, we know that  $f$  is  $\mu \times \nu$ -integrable iff  $I_{\|f\|}^-$  is  $\mu$ -integrable. Note that  $I_{\|f\|}^-$

## 24:10 Formalized Haar Measure

is similar to  $I_{\|f\|}$ . The difference is that when  $f_x$  is not integrable,  $I_{\|f\|}(x) = 0$  but  $I_{\|f\|}^- = \infty$ . This means that if  $I_{\|f\|}^-$  is  $\mu$ -integrable, then  $f_x$  must almost always be  $\nu$ -integrable (otherwise the integrand is infinite on a set of positive measure) and  $I_{\|f\|}$  must be  $\mu$ -integrable. For the other direction in Part 1, if  $f_x$  is almost always integrable, then  $I_{\|f\|} = I_{\|f\|}^-$  almost everywhere, so if the former is  $\mu$ -integrable, then so is the latter.

**Equality.** To prove the first equality in Part 2, we use the following *induction principle for integrable functions*.  $\square$  It states that to prove a statement  $P(f)$  for all  $\tau$ -integrable functions  $f : Z \rightarrow E$ , it is sufficient to show

- if  $A$  is measurable with  $\tau(A) < \infty$  and  $c \in E$ , then  $P(c\chi_A)$ ;
- if  $f_1$  and  $f_2$  are  $\tau$ -integrable functions with  $P(f_1)$  and  $P(f_2)$ , then  $P(f_1 + f_2)$ ;
- If  $f_1$  is integrable with  $P(f_1)$  and  $f_1 = f_2$   $\tau$ -a.e. then  $P(f_2)$ .
- The set  $\{f \in L^1(Z, \tau; E) \mid P(f)\}$  is closed.

We now prove by induction on  $f$  that

$$\int_{X \times Y} f \, d(\mu \times \nu) = \int_X \int_Y f(x, y) \, d\nu(y) d\mu(x). \quad (2)$$

- If  $f = c\chi_A$  then it is easy to see that both sides of (2) reduce to  $c \cdot (\mu \times \nu)(A)$ .
- It is clear that if (2) holds for  $f_1$  and  $f_2$ , then it also holds for  $f_1 + f_2$ .
- Suppose that (2) holds for  $f_1$  and that  $f_1 = f_2$  almost everywhere. Since integrals are defined up to a.e. equality, we know that

$$\int_{X \times Y} f_1 \, d(\mu \times \nu) = \int_{X \times Y} f_2 \, d(\mu \times \nu).$$

It therefore suffices to show that

$$\int_X \int_Y f_1(x, y) \, d\nu(y) d\mu(x) = \int_X \int_Y f_2(x, y) \, d\nu(y) d\mu(x).$$

For this it is sufficient to show that  $I_{f_1}(x) = I_{f_2}(x)$  for almost all  $x$ . Since  $f_1(z) = f_2(z)$  for almost all  $z$ , it is not hard to see  $\square$  that for almost all  $x$  we have that for almost all  $y$  the equality  $f_1(x, y) = f_2(x, y)$  holds. From this we easily get that  $I_{f_1}(x) = I_{f_2}(x)$  for almost all  $x$ .

- We need to show that the set of  $f \in L^1(Z, \tau; E)$  for which (2) holds is closed. We know that taking the integral of an  $L^1$  function is a continuous operation,  $\square$  which means that both sides of (2) are continuous in  $f$ , which means that the set where these functions are equal is closed.  $\blacktriangleleft$

## 5 Existence of the Haar Measure

In this section, we define the left Haar measure, and discuss some design decisions when formalizing these definitions. We also discuss how to obtain a right Haar measure.

Throughout this section, we assume that  $\mathbf{G}$  is a locally compact group, equipped with the Borel  $\sigma$ -algebra. We will write the group operation multiplicatively. We will define the Haar measure and show that it is a left invariant regular measure on  $\mathbf{G}$ . In the formalization, all intermediate definitions and results are given with weaker conditions on  $\mathbf{G}$ , following the design in `mathlib` that all results should be in the most general form (within reason).

In Lean, the precise conditions on  $\mathbf{G}$  are written as follows.

```
{G : Type*} [topological_space G] [t2_space G] [locally_compact_space G]
[group G] [measurable_space G] [borel_space G] [topological_group G]
```

The concept of a left invariant measure is defined as follows in `mathlib`. [↗](#)

```
def is_left_invariant (μ : set G → ennreal) : Prop :=
  ∀ (g : G) {A : set G} (h : is_measurable A), μ ((λ h, g * h) ⁻¹' A) = μ A
```

Note that the preimage  $(\lambda h, g * h)^{-1'} A$  is equal to the image  $(\lambda h, g^{-1} * h)' A$ . We use preimages for translations, since preimages are nicer to work with. For example, the fact that  $(\lambda h, g * h)^{-1'} A$  is measurable follows directly from the fact that multiplication (on the left) is a measurable function.

Next we give the definition of a regular measure [↗](#)

```
structure regular (μ : measure X) : Prop :=
  (lt_top_of_is_compact : ∀ {K : set X}, is_compact K → μ K < ⊤)
  (outer_regular : ∀ {A : set X}, is_measurable A →
    (⊓ (U : set X) (h : is_open U) (h2 : A ⊆ U), μ U) ≤ μ A)
  (inner_regular : ∀ {U : set X}, is_open U →
    μ U ≤ ⊔ (K : set X) (h : is_compact K) (h2 : K ⊆ U), μ K)
```

If  $\mu$  is a regular measure, then the inequalities in the last two fields are always equalities. This means that a measure  $\mu$  is regular if it is finite on compact sets, its value on a measurable set  $A$  is equal to the infimum ( $\bigcap$ ) of its value on all open sets containing  $A$ , and finally its value on an open set  $U$  is the supremum ( $\bigcup$ ) of its values on all compact subsets of  $U$ .

We are now ready to start the definition of the Haar measure. Given a compact set  $K$  in  $G$ , we start by giving a rough approximation of the “size” of  $K$  by comparing it to a reference set  $V$ , which is an open neighborhood of  $(1 : G)$ . We can consider all left translates  $(\lambda h, g * h)^{-1'} V$  of  $V$ , which is an open covering of  $K$ . Since  $K$  is compact, we only need finitely many left translates of  $V$  to cover  $K$ . [↗](#) Let  $\text{index } K \ V$  be the smallest number of left-translates of  $V$  needed to cover  $K$ . This is often denoted  $(K : V)$ . We did not use this in Lean code, since it conflicts with the typing-notation in Lean. In Lean, this notion is defined for arbitrary sets  $K$  and  $V$ , and it is defined to be 0 if there is no finite number of left translates covering  $K$ : [↗](#)

```
def index (K V : set G) : ℕ :=
  Inf (finset.card '' {t : finset G | K ⊆ ⋃ g ∈ t, (λ h, g * h) ⁻¹' V })
```

For the rest of this section, we fix  $K_0$  as an arbitrary compact set with nonempty interior (in `mathlib` this is denoted  $K_0 : \text{positive\_compacts } G$ ). We then define a weighted version of the index: [↗](#)

```
def prehaar (K_0 U : set G) (K : compact G) : ℝ :=
  (index K U : ℝ) / index K_0 U
```

This definition satisfies the following properties.

- **Lemma 4.** Denote  $\text{prehaar } K_0 \ U \ K$  by  $h_U(K)$  (recall that  $K_0$  is fixed). Then we have
- $(K : U) \leq (K : K_0) \cdot (K_0 : U)$ ; [↗](#)
  - $0 \leq h_U(K) \leq (K : K_0)$ ; [↗](#)
  - $h_U(K_0) = 1$ ; [↗](#)
  - $h_U(xK) = h_U(K)$ ; [↗](#)
  - if  $K \subseteq K'$  then  $h_U(K) \leq h_U(K')$ ; [↗](#)
  - $h_U(K \cup K') \leq h_U(K) + h_U(K')$  [↗](#) and equality holds if  $KU^{-1} \cap K'U^{-1} = \emptyset$ . [↗](#)

To define the left Haar measure, we next want to define the “limit” of this quotient as  $U$  becomes a smaller and smaller open neighborhood of 1. This is not an actual limit, but we

## 24:12 Formalized Haar Measure

will emulate it by constructing a product space that contains all these functions, and then use a compactness argument to find a “limit” in a large product space.

Consider the product of closed intervals

$$\prod_{K \subseteq G \text{ compact}} [0, (K : K_0)].$$

In Lean this is defined as a subspace of the topological space `compacts G → ℝ`:

```
def haar_product (K_0 : set G) : set (compacts G → ℝ) :=
  pi_univ (λ K : compacts G, Icc 0 (index K K_0))
```

Here `pi_univ` is the product of sets and `Icc` is an interval that is closed on both sides.

Note that by Tychonoff’s theorem `haar_product K_0` is compact, and that every function `prehaar K_0 U : compacts G → ℝ` determines a point in `haar_product K_0`.

Given an open neighborhood `V` of 1, we can define the collection of points determined by `prehaar K_0 U` for all `U ⊆ V` and take its closure:

```
def cl_prehaar (K_0 : set G) (V : open_nhds_of (1 : G)) :
  set (compacts G → ℝ) :=
  closure (prehaar K_0 " { U : set G | U ⊆ V ∧ is_open U ∧ (1 : G) ∈ U })
```

Now we claim that the intersection of `cl_prehaar K_0 V` is nonempty for all open neighborhoods `V` of 1.

```
lemma nonempty_Inter_cl_prehaar (K_0 : positive_compacts G) : nonempty
  (haar_product K_0 ∩ ⋂ (V : open_nhds_of (1 : G)), cl_prehaar K_0 V)
```

**Proof.** Since `haar_product K_0` is compact, it is sufficient to show that if we have a finite collection of open neighborhoods `t : finset (open_nhds_of 1)` the set `haar_product K_0 ∩ ⋂ (V ∈ t), cl_prehaar K_0 V` is non-empty. In this case we can explicitly give an element, namely `prehaar K_0 V_0`, where `V_0 = ⋂ (V ∈ t), V`. ◀

Finally, we can choose an arbitrary element in the set of the previous lemma, which we call `chaar K_0`, since it measures the size of the compact sets of `G`:

```
def chaar (K_0 : positive_compacts G) (K : compacts G) : ℝ :=
  classical.some (nonempty_Inter_cl_prehaar K_0) K
```

► **Lemma 5.** Denote `chaar K_0 K` by `h(K)`. Then we have

- $0 \leq h(K)$  and  $h(\emptyset) = 0$  and  $h(K_0) = 1$ ;
- $h(xK) = h(K)$ ;
- if  $K \subseteq K'$  then  $h(K) \leq h(K')$  (monotonicity);
- $h(K \cup K') \leq h(K) + h(K')$  (subadditivity) and equality holds if  $K \cap K' = \emptyset$  (additivity).

The idea in the proof is that the projections  $f \mapsto f(K)$  are continuous functions. Since these statements in this lemma hold for all  $h_U$  by Lemma 4, they also hold for each point in each set `cl_prehaar K_0 V`, and therefore for `h`.

This function `chaar K_0` measures the size of the compact sets of `G`. It has the properties of a *content* on the compact sets, which is an additive, subadditive and monotone function into the nonnegative reals [10, §53]. From this we can define the Haar measure in three steps.

First, given a content, we can obtain its associated *inner content* on the open sets,

```
def inner_content (h : compacts G → ennreal) (U : opens G) : ennreal :=
  ⋂ (K : compacts G) (hK : K ⊆ U), h K
```

This inner content is monotone and countably subadditive. Also note that for every set  $K$  that is both compact and open, we have  $\text{inner\_content } h K = h K$ .

Second, given an arbitrary inner content  $\nu$ , we can obtain its associated outer measure  $\mu$  which is given by  $\mu A = \inf \{ \sum (h U_i) \mid (U_i : \text{is\_open } U_i) (h : A \subseteq \bigcup U_i), \nu U_i \}$ .

For open sets  $U$  we have that  $\mu U = \nu U$ . Suppose that the inner content  $\nu$  comes from the content  $h$ . We have observed that  $\mu$  agrees with  $\nu$  on their common domain (the open sets), and that  $\nu$  agrees with  $h$  on their common domain (the compact open sets). One might be tempted to conclude that  $\mu$  agrees with  $h$  on their common domain (the compact sets). This is implicitly claimed in Gleason [8], since the same name is used for these three functions. However, this is not necessarily the case [10, p. 233]. The best we can say in general is that for compact  $K$  we have  $\mu (\text{interior } K) \leq h K \leq \mu K$ .

If we apply these two steps to the function  $h = \text{chaar } K_0$  we obtain the Haar outer measure  $\text{haar\_outer\_measure } K_0$  on  $G$ . Next, we show that all Borel measurable sets on  $G$  are Carathéodory measurable w.r.t.  $\text{haar\_outer\_measure } K_0$ . This gives the *left Haar measure*. We also scale the Haar measure so that it assigns measure 1 to  $K_0$ .

► **Theorem 6.** *Let  $\mu$  be the left Haar measure on  $G$ , defined above. Then  $\mu$  is a left invariant and regular measure satisfying  $\mu(K_0) = 1$ .*

We will show in Section 6 that the Haar measure is unique, up to multiplication determined by a constant. The following theorem states some properties of left invariant regular measures.

► **Theorem 7.** *Let  $\mu$  be a nonzero left invariant regular measure on  $G$ .*

- *If  $U$  is a nonempty open set, then  $\mu(U) > 0$ .*
- *If  $f : G \rightarrow \mathbb{R}$  is a nonnegative continuous function, not identically equal to 0, then  $\int f d\mu > 0$ .*

In this section we have only discussed the left Haar measure, which is left invariant. One can similarly define the notion of right invariance for a measure. We can easily translate between these two notions. For a measure  $\mu$  define the measure  $\check{\mu}$  by  $\check{\mu}(A) = \mu(A^{-1})$ . The following Theorem shows that if  $\mu$  is the left Haar measure on  $G$ , then  $\check{\mu}$  is a right Haar measure on  $G$  (i.e. a nonzero regular right invariant measure).

► **Theorem 8.**

- *The operation  $\mu \mapsto \check{\mu}$  is involutive.*
- *$\mu$  is left invariant iff  $\check{\mu}$  is right invariant.*
- *$\mu$  is regular iff  $\check{\mu}$  is regular.*

## 6 Uniqueness of the Haar measure

In this section we will show that the Haar measure is unique, up to multiplication by a constant. For this proof, we followed the proof in Halmos [10, §59-60] up to some differences given at the end of the section.

Let  $G$  be a second-countable locally compact group. The main ingredient in the proof is that if we have two left invariant measures  $\mu$  and  $\nu$  on  $G$ , we can transform expressions involving  $\mu$  into expressions involving  $\nu$  by applying a transformation to  $G \times G$  that preserves the measure  $\mu \times \nu$ . We need Tonelli's theorem to compute the integrals in this product. Because  $G$  is second countable, the product  $\sigma$ -algebra on  $G \times G$  coincides with the Borel  $\sigma$ -algebra.

Suppose  $\mu$  and  $\nu$  are left invariant and regular measures on  $G$ . Since  $G$  is second-countable, this also means that  $\mu$  and  $\nu$  are  $\sigma$ -finite. We first show that  $(x, y) \mapsto (x, xy)$  and  $(x, y) \mapsto (yx, x^{-1})$  are measure-preserving transformations.

## 24:14 Formalized Haar Measure

► **Lemma 9.** *Let  $\mu$  and  $\nu$  be left invariant regular measures on  $G$ , then the transformations  $S, T : G \times G \rightarrow G \times G$  given by  $S(x, y) := (x, xy)$  and  $T(x, y) := (yx, x^{-1})$  preserve the measure  $\mu \times \nu$ , i.e.  $S_*(\mu \times \nu) = \mu \times \nu = T_*(\mu \times \nu)$ .  $\square$   $\square$*

**Proof.** It suffices to show that the measures are equal on measurable rectangles, since these generate the  $\sigma$ -algebra on  $G \times G$ . Let  $A \times B \subseteq G \times G$  be a measurable rectangle. For  $S$  we compute:

$$\begin{aligned} S_*(\mu \times \nu)(A \times B) &= (\mu \times \nu)(S^{-1}(A \times B)) \\ &= \int^- \nu(S(x, -)^{-1}(A \times B)) \, d\mu(x) = \int^- \chi_A(x) \nu(x^{-1}B) \, d\mu(x) \\ &= \int^- \chi_A(x) \nu(B) \, d\mu(x) = \mu(A) \nu(B) = (\mu \times \nu)(A \times B). \end{aligned}$$

Note that  $S$  is an equivalence with inverse  $S^{-1}(x, y) = (x, x^{-1}y)$ . If we define  $R(x, y) := (y, x)$ , we saw in Section 4 that  $R_*(\mu \times \nu) = \nu \times \mu$ . The claim that  $T$  preserves measure now follows from the observation that  $T = S^{-1}R$ , which is easy to check.  $\blacktriangleleft$

For measurable sets  $A$  its left translates have the same measure:  $\mu(xA) = \mu(A)$ . This might not be true for right translates  $\mu(Ax)$ , but we can say the following [10, §59, Th. D].

► **Lemma 10.** *Let  $\mu$  be a left invariant regular measure on  $G$  and suppose that  $\mu(A) > 0$  for a measurable set  $A \subseteq G$ . Then  $\mu(A^{-1}) > 0$   $\square$  and  $\mu(Ax) > 0$   $\square$  for  $x \in G$ . Furthermore, the map  $f(x) := \mu(Ax)$  is measurable.  $\square$*

**Proof.** Let  $S$  and  $T$  be as in Lemma 9. For the first claim, we will show that  $\mu(A^{-1}) = 0$  implies that  $\mu(A) = 0$ . It suffices to show that  $(\mu \times \mu)(A \times A) = 0$ , which we can compute using Lemma 9:

$$\begin{aligned} (\mu \times \mu)(A \times A) &= (\mu \times \mu)(T^{-1}(A \times A)) = \int^- \mu((T^{-1}(A \times A))^y) \, d\mu(y) \\ &= \int^- \mu(y^{-1}A \cap A^{-1}) \, d\mu(y) \leq \int^- \mu(A^{-1}) \, d\mu(y) = 0. \end{aligned}$$

The second claim now easily follows, since  $\mu(A) > 0$  implies that  $\mu(Ax) = \mu((x^{-1}A^{-1})^{-1}) > 0$ .

For the final claim, we can see that  $\mu(Ax) = \mu((S^{-1}(G \times A))^x)$ , which is measurable by the symmetric version of `measurable_measure_prod_mk_left` from Section 4.  $\blacktriangleleft$

The following is a technical lemma that allows us to rewrite  $\nu$ -integrals into  $\mu$ -integrals. This is the key lemma to show the uniqueness of the Haar measure. [10, §60, Th. A].

► **Lemma 11.** *Let  $\mu$  and  $\nu$  be left invariant regular measures on  $G$ , let  $K \subseteq G$  be compact with  $\nu(K) > 0$ , and suppose that  $f : G \rightarrow [0, \infty]$  is measurable. Then  $\square$*

$$\mu(K) \cdot \int^- \frac{f(y^{-1})}{\nu(Ky)} \, d\nu(y) = \int^- f \, d\mu.$$

**Proof.** First note that  $0 < \nu(Ky) < \infty$  for any  $y$ . The first inequality follows from Lemma 10 and the second inequality from the fact that  $\nu$  is regular and  $Ky$  is compact. Let





## 24:16 Formalized Haar Measure

proof of the Haar measure, to the calculational proofs in the uniqueness of the Haar measure. This formalization shows that these different proof techniques can be used and combined effectively in `mathlib` to formalize graduate level mathematics.

This work can be extended in various ways. One interesting project would be to prove the uniqueness for groups that need not be second countable. This should not be too difficult, but requires a version of Fubini’s theorem for compactly supported functions on locally compact Hausdorff spaces. Another project would be to explore the *modular function*, which describes how the left and the right Haar measures differ on a locally compact group. Furthermore, the Haar measure is used as a tool in many areas of mathematics, some of which would now be feasible to formalize with the Haar measure as a new tool in our toolkit.

One such area is abstract harmonic analysis. Abstract harmonic analysis is the area of analysis on topological groups, usually locally compact groups. One of the core ideas is to generalize the Fourier transform to an arbitrary locally compact abelian group. The special case of the continuous Fourier transform has been formalized in HOL4 [9], but the abstract Fourier transform has never been formalized. One specific goal could be to formalize *Pontryagin duality*. For a locally compact abelian group  $G$  we can define the *Pontryagin dual*  $\widehat{G} := \text{Hom}(G, T)$  consisting of the continuous group homomorphisms from  $G$  to the circle group  $T$ . Then Pontryagin duality states that the canonical map  $G \rightarrow \widehat{\widehat{G}}$  (given by the evaluation function) is an isomorphism. Other targets include Peter–Weyl theorem and more ambitiously the representation theory of compact groups and Weyl’s integral formula for compact Lie groups.

In `mathlib` we have not just defined binary product measures, but also finitary product measures,  $\prod$  which can be used to define integrals over  $\mathbb{R}^n$ . In `mathlib` this is done by defining a measure on the product space  $\prod i : \iota, X i$ .

```
def measure.pi {ι : Type*} [fintype ι] {X : ι → Type*}
  [∀ i, measurable_space (X i)] (μ : ∀ i, measure (X i)) :
  measure (∏ i, X i)
```

The definition is conceptually very simple, since you just iterate the construction of binary product measures, but in practice it is quite tricky to do it in a way that is convenient to use. For example, in a complicated proof, you do not want to worry about the fact that the spaces  $X \times (Y \times Z)$  and  $(X \times Y) \times Z$  are not exactly the same, just equivalent. One open question is whether we can formulate Fubini’s theorem for finitary product measures in a way that is convenient to use. It would be inconvenient to rewrite a finitary product of measures into a binary product in the desired way, and then apply Fubini’s theorem for binary product measures. We have an idea that we expect to be more promising, which is to define a “marginal integral” that only integrates some variables in the input of a function, corresponding to the marginal distribution of a random variable in probability theory. It takes a function  $f : (\prod i, X i) \rightarrow E$  and integrates away the variables in a subset  $I$ . The result is another function  $(\prod i, X i) \rightarrow E$  that is constant on all variables in  $I$ :

```
def marginal (μ : ∀ i, measure (X i)) (I : finset ι)
  (f : (∏ i, X i) → E) : (∏ i, X i) → E
```

This has not been incorporated in `mathlib`, but it is a promising experiment on a separate branch.  $\square$  If we denote  $\text{marginal } \mu \ I \ f$  as  $\int \cdots \int_{-I} f \ \partial \mu$ , then the following is a convenient way to state Fubini’s theorem for finitary product measures.

```
lemma marginal_union (f : (∏ i, X i) → E) (I J : finset ι) :
  disjoint I J →  $\int \cdots \int_{-I \cup J} f \ \partial \mu = \int \cdots \int_{-I} f \ \partial \mu \cdot \int \cdots \int_{-J} f \ \partial \mu$ 
```

## References

- 1 Jeremy Avigad, Johannes Hölzl, and Luke Serafin. A formally verified proof of the central limit theorem. *Journal of Automated Reasoning*, 59(4):389–423, 2017.
- 2 Aaron R. Coble. Anonymity, information, and machine-assisted proof. Technical Report UCAM-CL-TR-785, University of Cambridge, Computer Laboratory, July 2010. URL: <https://www.cl.cam.ac.uk/techreports/UCAM-CL-TR-785.pdf>.
- 3 Donald L. Cohn. *Measure Theory*. Birkhäuser Basel, 2 edition, 2013.
- 4 The mathlib Community. The Lean mathematical library. In *Proceedings of the 9th ACM SIGPLAN International Conference on Certified Programs and Proofs, CPP 2020*, page 367–381, New York, NY, USA, 2020. Association for Computing Machinery. doi:10.1145/3372885.3373824.
- 5 Leonardo de Moura, Soonho Kong, Jeremy Avigad, Floris van Doorn, and Jakob von Raumer. The Lean Theorem Prover (system description). *CADE-25*, pages 378–388, August 2015.
- 6 Gabriel Ebner, Sebastian Ullrich, Jared Roesch, Jeremy Avigad, and Leonardo de Moura. A metaprogramming framework for formal verification. *Proc. ACM Program. Lang.*, 1(ICFP):34:1–34:29, aug 2017. doi:10.1145/3110278.
- 7 Noboru Endou, Keiko Narita, and Yasunari Shidama. The Lebesgue monotone convergence theorem. *Formalized Mathematics*, 16(2):167–175, 2008.
- 8 Jonathan Gleason. Existence and uniqueness of Haar measure. preprint, 2010.
- 9 Yong Guan, Jie Zhang, Zhiping Shi, Yi Wang, and Yongdong Li. Formalization of continuous Fourier transform in verifying applications for dependable cyber-physical systems. *Journal of Systems Architecture*, 106:101707, 2020. URL: <http://www.sciencedirect.com/science/article/pii/S1383762120300011>, doi:<https://doi.org/10.1016/j.sysarc.2020.101707>.
- 10 Paul R. Halmos. *Measure theory*. Springer-Verlag New York, 1950. doi:10.1007/978-1-4684-9440-2.
- 11 John Harrison. The HOL Light theory of Euclidean space. *Journal of Automated Reasoning*, 50(2):173–190, 2013. doi:10.1007/s10817-012-9250-9.
- 12 Johannes Hölzl and Armin Heller. Three chapters of measure theory in Isabelle/HOL. In *Interactive Theorem Proving*, pages 135–151, Berlin, Heidelberg, 2011. Springer Berlin Heidelberg.
- 13 Joe Hurd. Formal verification of probabilistic algorithms. Technical Report UCAM-CL-TR-566, University of Cambridge, Computer Laboratory, May 2003. URL: <https://www.cl.cam.ac.uk/techreports/UCAM-CL-TR-566.pdf>.
- 14 David R Lester. Topology in PVS: Continuous mathematics with applications. In *Proceedings of the Second Workshop on Automated Formal Methods, AFM '07*, page 11–20, New York, NY, USA, 2007. Association for Computing Machinery. doi:10.1145/1345169.1345171.
- 15 Tarek Mhamdi, Osman Hasan, and Sofiène Tahar. Formalization of measure theory and Lebesgue integration for probabilistic analysis in HOL. *ACM Trans. Embed. Comput. Syst.*, 12(1), January 2013. doi:10.1145/2406336.2406349.
- 16 Jan Mikusiński. The bochner integral. In *The Bochner Integral*, pages 15–22. Springer, 1978.
- 17 Stefan Richter. Formalizing integration theory with an application to probabilistic algorithms. In *Theorem Proving in Higher Order Logics*, pages 271–286, Berlin, Heidelberg, 2004. Springer Berlin Heidelberg.
- 18 B.A.W. Spitters and E.E. van der Weegen. Type classes for mathematics in type theory. *Mathematical Structures in Computer Science*, 21:795–825, 2011.
- 19 Joseph Tassarotti. coq-proba probability library, 2019. URL: <https://github.com/jtassarotti/coq-proba>.
- 20 P. Wadler and S. Blott. How to make ad-hoc polymorphism less ad hoc. In *Proceedings of the 16th ACM SIGPLAN-SIGACT Symposium on Principles of Programming Languages, POPL '89*, page 60–76, New York, NY, USA, 1989. Association for Computing Machinery. URL: <https://doi.org/10.1145/75277.75283>, doi:10.1145/75277.75283.